


ARMY RESEARCH LABORATORY



Helicopter Trim Analysis

Joseph Fries

ARL-MR-317

June 1996

APPROVED FOR PUBLIC RELEASE; DISTRIBUTION IS UNLIMITED.

19960708 036

DTIC QUALITY INSPECTED 1

NOTICES

Destroy this report when it is no longer needed. DO NOT return it to the originator.

Additional copies of this report may be obtained from the National Technical Information Service, U.S. Department of Commerce, 5285 Port Royal Road, Springfield, VA 22161.

The findings of this report are not to be construed as an official Department of the Army position, unless so designated by other authorized documents.

The use of trade names or manufacturers' names in this report does not constitute indorsement of any commercial product.

REPORT DOCUMENTATION PAGE			Form Approved OMB No. 0704-0188	
<small>Public reporting burden for this collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302, and to the Office of Management and Budget, Paperwork Reduction Project(0704-0188), Washington, DC 20503.</small>				
1. AGENCY USE ONLY (Leave blank)		2. REPORT DATE June 1996		3. REPORT TYPE AND DATES COVERED Final, 10-15 Sep 92
4. TITLE AND SUBTITLE Helicopter Trim Analysis			5. FUNDING NUMBERS WO: 44592-202-94-2123 PR: 1L162618AH80	
6. AUTHOR(S) Joseph Fries				
7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES) U.S. Army Research Laboratory ATTN: AMSRL-SL-BA Aberdeen Proving Ground, MD 21005-5068			8. PERFORMING ORGANIZATION REPORT NUMBER ARL-MR-317	
9. SPONSORING/MONITORING AGENCY NAMES(S) AND ADDRESS(ES)			10. SPONSORING/MONITORING AGENCY REPORT NUMBER	
11. SUPPLEMENTARY NOTES				
12a. DISTRIBUTION/AVAILABILITY STATEMENT Approved for public release; distribution is unlimited.			12b. DISTRIBUTION CODE	
13. ABSTRACT (Maximum 200 words) This report presents a generic first-principles helicopter trim analysis for use on a personal work station. Working from the helicopter's physical properties, it calculates the main rotor blade's collective, longitudinal cyclic, body pitch, and stabilator pitch angles to maintain a trimmed flight condition over a complete forward-speed range. The computer program automatically sweeps through a forward-speed range, varying the collective, longitudinal cyclic, and stabilator pitch angles to satisfy the trimmed condition. It is an improvement over earlier work where the analyst had to vary these parameters at the keyboard. It presents an example of the usage with an AH-64A helicopter, and the program's source code is included.				
14. SUBJECT TERMS trim, helicopter, stead-state flight, flight dynamics, stability and control			15. NUMBER OF PAGES 31	
			16. PRICE CODE	
17. SECURITY CLASSIFICATION OF REPORT UNCLASSIFIED	18. SECURITY CLASSIFICATION OF THIS PAGE UNCLASSIFIED	19. SECURITY CLASSIFICATION OF ABSTRACT UNCLASSIFIED	20. LIMITATION OF ABSTRACT UL	

INTENTIONALLY LEFT BLANK.

PREFACE

The preliminary results of an effort to develop a tool for determining trim flight control settings (expressed as aerodynamic surface pitch angles) for a helicopter with vertical, horizontal, and pitch degrees of freedom are presented. This task was initiated to produce a faster, less labor-intensive method to compute control settings (i.e., initial condition inputs) for a damaged helicopter descent dynamics model used in conjunction with ballistic vulnerability analyses. Previously, a time-consuming manual iteration approach was used to determine the needed values.

A method is described whereby trim is accomplished by optimizing the main rotor blade collective pitch and longitudinal cyclic pitch settings and the pitch angle of the horizontal stabilizer to achieve balanced forces and moments and minimum required power for a given airspeed. Main rotor rotational and individual blade flapping degrees of freedom are also included in the model, which is set up for conventional single main rotor helicopter configurations. The computational procedure, implemented in a BASIC language computer program, determines trim control settings for steady, level flight (i.e., at constant forward speed, constant altitude, zero aircraft pitch velocity, and constant rotor speed within the range of available power). A program feature allows the trim solution, a function of forward airspeed, to be determined over a range of velocities with a single execution run.

An application example (AH-64A Apache) with numerical results is given, including an input data set. A program listing is contained in an appendix. The information in this report allows use of the program and, as well, is a foundation for further development and verification of a prototype procedure.

INTENTIONALLY LEFT BLANK.

TABLE OF CONTENTS

	<u>Page</u>
PREFACE	iii
LIST OF FIGURES	vii
LIST OF EQUATIONS	vii
1. INTRODUCTION	1
2. ANALYTICAL APPROACH	1
3. APPLICATION EXAMPLE	3
4. CONCLUSION	9
5. REFERENCES	11
APPENDIX A: INPUT DATA SET - AH-64A	13
APPENDIX B: PROGRAM LISTING - BASIC	17
LIST OF ABBREVIATIONS AND SYMBOLS	29
DISTRIBUTION LIST	31

INTENTIONALLY LEFT BLANK.

LIST OF FIGURES

<u>Figure</u>	<u>Page</u>
1. Collective pitch (deg.) vs. forward speed (knots)	4
2. Longitudinal cyclic pitch (deg.) vs. forward speed (knots)	4
3. Aircraft pitch angle (deg.) vs. forward airspeed (knots)	5
4. Stabilator pitch angle (deg.) vs. forward airspeed (knots)	5
5. Required power (HP) vs. forward airspeed (knots)	6
6. Main rotor speed (RPM) vs. forward airspeed (knots)	6
7. Disc-plane pitch angle (deg.) vs. forward airspeed (knots)	7
8. Collective pitch (deg.) vs. forward speed (knots) (closed form solution [CFS])	7
9. Longitudinal cyclic pitch angle (deg.) vs. forward airspeed (knots) (CFS)	8
10. Required power (HP) vs. forward airspeed (knots) (CFS)	8

LIST OF EQUATIONS

<u>Equations</u>	<u>Page</u>
1. General Solution Equation for Displacement Motion	2
2. Basic Equation for Aerodynamic Lift Force	2
3. Basic Equation for Aerodynamic Drag Force	2

INTENTIONALLY LEFT BLANK.

1. INTRODUCTION

The Air Systems Branch (ASB) of the Ballistic Vulnerability/Lethality Division (BVLD), Survivability/Lethality Analysis Directorate (SLAD), U.S. Army Research Laboratory (ARL), is tasked with evaluating the vulnerability of helicopter systems to ballistic threats (e.g., armor piercing and high-explosive projectiles). A component of these evaluations involves determining the aircraft's response to changes in engine output power caused by ballistic damage. These assessments, presented in the form of height-velocity dependent kill boundaries, are made using an equations-of-motion model that calculates the aircraft's trajectory from a trimmed flight state following loss of engine power (see Helicopter Descent Computer Model [U.S. Army Ballistic Research Laboratory 1990]). This model requires specification of the main rotor control settings for trimmed level flight as an initial condition.

This report presents a prototype method for determining trim flight control settings for a conventional single main rotor helicopter (i.e., main/tail rotor configuration) with vertical, horizontal, and pitch degrees of freedom. Trim is accomplished by optimizing the main rotor blade collective pitch and longitudinal cyclic pitch settings and the pitch angle of the horizontal stabilizer (if fitted) to achieve balanced forces and moments and minimum power required for a given airspeed. Main rotor rotational and individual blade flapping degrees of freedom are also included in the model.

The computerized procedure calculates the trim settings for steady-level flight (i.e., for constant forward speed, constant altitude, zero aircraft pitch velocity, and constant rotor speed within the range of available power). A program feature allows the trim solution, a function of forward airspeed, to be determined over a range of velocities with a single execution run.

2. ANALYTICAL APPROACH

The computational procedure models a helicopter having three degrees of helicopter body freedom (rectilinear motion along the longitudinal and vertical axes, and angular pitch about the lateral axis) along with additional main rotor rotational motion and "n" blade flapping degrees of freedom dependent on the number of main rotor blades. A four-blade main rotor, as an example, gives a system with a total of eight degrees of freedom with this model. The system's dynamics are expressed with a simplified set of displacement equations derived from the general equations of motion. The basic form of the solution

equation for displacement motion is shown in equation 1. (Note: equation terms are defined in the Abbreviations and Symbols Section at the end of this report.)

$$X(2) = X(1) + (dX(1)/dt) \delta t + (d^2X(1)/dt^2) (\delta t^2/2). \quad (1)$$

General Solution Equation for Displacement Motion

The mathematical solution for the displacement equation set involves a Taylor series time step approach (see Riddle 1974). With this numerical representation, present displacements are functions of the velocities and the accelerations of the previous time step. The time step interval, an input variable (DS, see Appendix A), is expressed in terms of degrees of main rotor rotation, with the main rotor rotational speed (RPM) providing the time reference.

Main rotor and stabilator aerodynamic loads, lift and drag, force the dynamic system. The basic lift and drag equations are shown in equations 2 and 3, respectively. Blade element theory is used to estimate the main rotor loads (see Johnson 1980). Individual blade aerodynamic forces are calculated by integrating the contributions of discrete spanwise segments of the blade from the blade root to the tip. The contributions of each blade (i.e., for "n" blades) are then summed instantaneously in time to give the resultant total force acting at the rotor center during that time step.

$$dL/dr = 0.5 \rho c a \alpha (U_p^2 + U_t^2). \quad (2)$$

Basic Equation for Aerodynamic Lift Force

$$dD/dr = 0.5 \rho c C_d (U_p^2 + U_t^2). \quad (3)$$

Basic Equation for Aerodynamic Drag Force

Trim is satisfied when computed aircraft longitudinal acceleration, vertical velocity, and pitch velocity, after 10 rotor revolutions with fixed values of collective and longitudinal cyclic pitch and stabilator pitch angle, are equal to or less than specified criteria of $\pm 5 \text{ ft/s}^2$, $\pm 5 \text{ ft/s}$, and $2^\circ/\text{s}$, respectively. These criteria may be easily tightened/relaxed by changing the source code's numerical values. Aircraft (fuselage) pitch is held at a computed minimum drag attitude. At each specified airspeed, the three control settings (collective, longitudinal cyclic, and stabilator pitch angles) are varied within the program by ± 0.1 or $\pm 0.2^\circ$

until the criteria for residual accelerations and velocities are satisfied. Rotor collective and stabilator pitch are each defined as positive nose up. Longitudinal cyclic pitch is defined as positive nose up when the rotor blade's azimuthal position is 90° off the aircraft nose in the direction of blade rotation looking down from above the rotor. The computational procedures are implemented in a BASIC-language computer code (see Appendix B).

3. APPLICATION EXAMPLE

Application of the program was made for the AH-64A Apache helicopter. The corresponding input data file is given in Appendix A. Output results—collective pitch, longitudinal cyclic pitch, aircraft pitch, stabilizer pitch, required power and disk plane attitude, all as a function of forward speed—are given in Figures 1–6, respectively. For comparison, select terms were also computed manually with simplified closed form dynamic equilibrium equations. These equations were written to satisfy conditions of constant forward speed and zero vertical velocity without including aircraft longitudinal pitch or stabilator effects. Their solution yielded trends for collective pitch, longitudinal cyclic pitch, and required power as a function of forward speed. For parametric comparison, three values of main rotor blade lift curve slope were evaluated. Results are shown in Figures 7–9, respectively. The variations in each term with airspeed are illustrative; the computed values are approximate (due to the simplified form of the equations); however, the shapes of the curves compare favorably to textbook examples (see Chopra 1991).

In general, the computer model results/trends were comparable to the manual estimates and the literature examples. There was one exception, however; significant disparity existed between the computer model and manual (Figures 2 and 9) values for longitudinal cyclic pitch. The model indicated a positive reversal in longitudinal cyclic pitch beginning at 40 knots (i.e., at 40 knots the model required positive increasing longitudinal cyclic pitch with airspeed in order to trim), conflicting with the manual estimates and flight test measurements (U.S. Army Aviation Engineering Flight Activity 1982), which showed sustained longitudinal cyclic pitch reduction with increasing forward speed to trim. The discrepancy was attributed to a fault in the representation of the stabilator pitch control schedule and/or control power in the program. While the curve's shape (Figure 9) was appropriate, the pitch angle's absolute values were too great, particularly as airspeed increases. The positive increasing longitudinal cyclic pitch (Figure 2) was needed to balance the excessive stabilator forces and avoid extreme fuselage pitch attitudes. Figure 10 shows the required power trends for three different lift slope values.




Figure 1. Collective pitch (deg.) vs. forward speed (knots).




Figure 2. Longitudinal cyclic pitch (deg.) vs. forward speed (knots).




Figure 3. Aircraft pitch angle (deg.) vs. forward airspeed (knots).




Figure 4. Stabilator pitch angle (deg.) vs. forward airspeed (knots).




Figure 5. Required power (HP) vs. forward airspeed (knots).




Figure 6. Main rotor speed (RPM) vs. forward airspeed (knots).




Figure 7. Disc-plane pitch angle (deg.) vs. forward airspeed (knots).




Figure 8. Collective pitch (deg.) vs. forward speed (knots) [CFS].




Figure 9. Longitudinal pitch angle (deg.) vs. forward airspeed (knots) [CFS].




Figure 10. Required power (HP) vs. forward airspeed (knots) [CFS].

4. CONCLUSION

This report presents the initial results/findings of an effort to develop a computational tool for determining helicopter trim flight control settings. The aircraft is modeled as a system having three degrees of helicopter body freedom (rectilinear motion along the longitudinal and vertical axes, and angular pitch about the lateral axis) along with main rotor rotational motion and "n" blade flapping degrees of freedom. System dynamics are expressed with a set of displacement equations derived from the general equations of motion. These are solved numerically with a Taylor series time step approach. A computer code, written to facilitate the computational process, is described along with an input data set for the AH-64A. Results via the computer model are compared with (baseline) information in the literature and with estimates calculated manually with closed form dynamic equations.

The comparison indicated general agreement between the values produced with the computer model and the baseline values with the exception of longitudinal cyclic pitch. That difference was attributed to problems with the representations of the stabilator control schedule/power as a function of airspeed in the program. This phase of the effort has yielded an improved understanding of helicopter flight dynamics and a prototype engineering tool which, compared with earlier in-house methods, more efficiently computes desired trim control settings. Further development and application of this program is planned in conjunction with analyses of helicopter response to engine power loss and autorotation studies.

INTENTIONALLY LEFT BLANK.

5. REFERENCES

Chopra, I. "Notes on Helicopter Dynamics." Ch. 1, Dept. of Aerospace Engineering, University of Maryland, College Park, MD, January 1991.

Johnson, W. Helicopter Theory. Princeton, NJ: Princeton University Press, p. 51, 1980.

Riddle, D. F. Calculus and Analytic Geometry. 2nd edition, Belmont, CA: Wadsworth Publishing Co., p. 599, 1974.

U.S. Army Aviation Engineering Flight Activity. "Airworthiness and Flight Characteristics (A&FC) Test of YAH-64 Advanced Attack Helicopter, Prototype Qualification Test-Government (PQT-G), Part 3 and Production Validation Test-Government (PVT-G) for Handbook Verification." Final Report, USAAEFA Project 80-17-3, Edwards Air Force Base, CA, October 1982.

U.S. Army Ballistic Research Laboratory. "Helicopter Descent Computer Model." Unpublished work, Ballistic Vulnerability/Lethality Directorate, Air Systems Branch, Aberdeen Proving Ground, MD, 1990.

INTENTIONALLY LEFT BLANK.

APPENDIX A:
INPUT DATA SET - AH-64A

INTENTIONALLY LEFT BLANK.

Input Data for Computer code pls2
Located in Director pwrloss on NORTHGATE PC

VARIABLE	VALUE	DEFINITION
GA	-5	Shaft tilt, degrees (- into wind)
ZR	8.33	Main rotor height, ft and above CG
CG	16	CG, in (and forward of main rotor shaft)
XT	28.8	Distance from CG to stabilizer, ft (+ aft of CG)
ZT	0	Height of stabilizer above, CG, ft and up
TW	-0.375	Main rotor geometric twist, deg/ft and nose up
RM	24	Main rotor tip radius, ft
RT	5.58	Stabilizer semi-span, ft
TO	—	Main rotor collective pitch, deg
T2	—	Main rotor longitudinal cyclic pitch, deg and nose up at PSI = 90° azimuth
RH	0.002378	Air density slug/cubic ft
CM	21	Main rotor chord, in
CT	31.85	Stabilizer chord, in
MM	6	Main rotor blade lift slope, ND
TM	6	Stabilizer lift slope, ND
OT	—	Stabilizer pitch, deg nose up
EM	0	Offset angle of attack of main rotor, deg
ET	0	Offset angle of attack of stabilizer, deg
DO	0.014	Main rotor steady drag coefficient, ND
OD	0.014	Stabilizer steady drag coefficient, ND
K	0	Main rotor high order drag coefficient, ND
KT	0	Stabilizer high order drag coefficient, ND
PP	—	Initial aircraft pitch attitude, deg (+ nose up)
WB	166	Blade weight, lb
AP	2,828	Available power, HP
WT	16,000	Aircraft gross weight, lb
LL	13.5	Aircraft effective length, ft used to calculate effective frontal area
WW	6	Aircraft effective width, ft used to calculate effective frontal area
HH	5.03	Aircraft effective height, ft used to calculate effective front area

V	—	Aircraft forward speed, knots
FT	—	Final computer run time, sec
DS	15	Azimuthal increment of computer calculations, deg
OM	289.146	Rotor speed, RPM
AL	—	Aircraft altitude, ft
NB	4	Number of blades in main rotor, ND
NE	10	Number of blade elements for radial integration, ND
OB	1.0	Blade 1st flap frequency ratio, ND
TK	1	Main rotor blade thickness, in
CC	0.0125	Numerical damping ratio, ND
EE	0.05	Convergence factor
FA	—	Speed sweep factor \pm
NC	—	Number of rotor cycles to run, ND

APPENDIX B:
PROGRAM LISTING - BASIC

INTENTIONALLY LEFT BLANK.

```

10 G$="pls2 09/03/92 13:00"
20 OPEN "I",1,"APACHE"
30 OPEN "O",2,"OUT"
40 REM"....."
50 REM".....INPUT DATA....."
60 REM"....."
70 INPUT#1,GA,ZR,CG,XT,ZT
80 PRINT GA;ZR;CG;XT;ZT
90 PRINT#2, "GA=";GA;"SHAFT TILT DEG - DOWN FROM VERT INTO WIND"
100 PRINT#2, "ZR=";ZR;"MAIN ROTOR HT + ABOVE CG FT"
110 PRINT#2, "CG=";CG;"CG + FWD OF MAIN ROTOR SHAFT, IN"
120 PRINT#2, "XT=";XT;"DISTANCE FROM CG TO STAB, FT + AFT OF CG"
130 PRINT#2, "ZT=";ZT;"HT OF STAB ABOVE CG + UP FT"
140 REM GA=Shaft Tilt,Deg,-Into Wind
150 REM ZR=Main Rotor Ht,Ft,+Above CG
160 REM CG=CG Position,Inches,+ Fwd Of Main Rotor Shaft
170 REM Do Not Set CG=0!
180 REM XT=Distance From CG To Stab,Ft,+ Aft From CG
190 REM ZT=Height OF Stab Above CG,Ft,+ Up
200 INPUT#1,TW,RM,RT,T0,T2
210 PRINT TW;RM;RT;T0;T2
220 PRINT#2, "TW=";TW;"MAIN ROTOR GEOMETRIC TWIST DEG/FT + NOSE UP"
230 PRINT#2, "RM=";RM;"MAIN ROTOR TIP RADIUS, FT"
240 PRINT#2, "RT=";RT;"STAB RADIUS OR SEMI SPAN, FT"
250 PRINT#2, "t0=";t0;"MAIN ROTOR COLLECTIVE PITCH, DEG + NOSE UP"
260 PRINT#2, "T2=";T2;"MAIN ROTOR LONG CYCLIC PITCH,DEG + NOSE UP@PSI=90"
270 REM T0=Main Rotor Collective Pitch,Deg,+Nose Up
280 REM TW=Main Rotor Geometric Twist,Deg/Ft,+Nose Up
290 REM RM=Main Rotor Radius,Ft
300 REM RT=Stab Radius Or Semi-Span,Ft
310 REM T2=Main Rotor Long.Cyclic Pitch,Deg,+Nose Up At PSI=90
320 INPUT#1,RH,CM,CT,MM,TM
330 PRINT RH;CM;CT;MM;TM
340 PRINT#2, "RH=";RH;"AIR DENSITY,SLUG/FT3"
350 PRINT#2, "CM=";CM;"MAIN ROTOR CHORD,IN"
360 PRINT#2, "CT=";CT;"STAB CHORD,IN"
370 PRINT#2, "MM=";MM;"dCL/dALPHA MAX OF MAIN ROTOR BLADE,ND"
380 PRINT#2, "TM=";TM;"dCL/dALPHA MAX OF STAB,ND"
390 REM RH=Air Density,Slug/Ft3
400 REM CM=Main Rotor Blade Chord,Inches
410 REM CT=Stab Chord,Inches
420 REM MM=dCL/dALPHA Max Of Main Rotor,ND
430 REM TM=dCL/dALPHA Max of Stab,ND
440 INPUT#1,OT,EM,ET,DO,OD
450 PRINT OT;EM;ET;DO;OD
460 PRINT#2, "OT=";OT;"STAB PITCH ANGLE,DEG + NOSE UP"
470 PRINT#2, "EM=";EM;"1/2 SIN-1(CL OFFSET @ 0 ANGLE OF ATTACK/CL MAX)MAIN ROTOR,ND"
480 PRINT#2, "ET=";ET;" SAME BUT FOR STAB "
490 PRINT#2, "DO=";DO;"STEADY DRAG COEFF MAIN ROTOR, ND"
500 PRINT#2, "OD=";OD;"STEADY DRAG COEFF STAB,ND"
510 REM OT=Stab Pitch Angle,Deg,+ Nose Up
520 REM EM=1/2 Sin-1(CL Offset @ 0 Angle Of Attack/CL Max),Main Rotor,ND
530 REM ET=Same As EM, But For Stab
540 REM DO=Steady Drag Coeff.,Main Rotor,ND
550 REM OD= " " " ,Stab,ND
560 INPUT#1,K,KT,PP,WB,AP
570 PRINT K;KT;PP;WB;AP
580 PRINT#2, "K=";KT;"DRAG COEFF OF MAIN ROTOR SIN 2 TERM,ND"
590 PRINT#2, "KT=";KT;"DRAG COEFF OF STAB SIN 2 TERM,ND"
600 PRINT#2, "PP=";PP;"INITIAL AC PITCH,DEG + NOSE UP"
610 PRINT#2, "WB=";WB;"MAIN ROTOR BLADE WT, LB"
620 PRINT#2, "AP=";AP;"AVAILABLE ENGINE POWER,HP"
630 REM K+Drag coeff, Of Main Rotor Sin2 Term,ND

```

```

640 REM KT="      "      "      Stab Sin2 Term,ND
650 REM PP=Initial Aircraft Pitch Angle,Deg,+Nose Up
660 REM WB=Main Rotor Blade Wt,Lb
670 REM AP=Available Engine Power,HP
680 INPUT#1,WT,LL,WW,HH
690 PRINT WT;LL;WW;HH
700 PRINT#2,"T=";WT;"AIRCRAFT GROSS WT,LB"
710 PRINT#2,"LL=";LL;"AIRCRAFT EFF LENGTH FOR DRAG,FT"
720 PRINT#2,"WW=";WW;"AIRCRAFT EFF WIDTH FOR DRAG,FT"
730 PRINT#2,"HH=";HH;"AIRCRAFT EFF HEIGHT FOR DRAG,FT"
740 REM WT=Aircraft Wt,Lb
750 REM LL=Aircraft Effective Length, For Drag Calc.,Ft
760 REM WW=Aircraft Effective Width, "      "      "      "
770 REM HH=Aircraft Effective Height, "      "      "      "
780 INPUT#1,V,FT,DS,OM,AL
790 PRINT V;FT;DS;OM;AL
800 PRINT#2,"V=";V;"AIRCRAFT SPEED,KT"
810 PRINT#2,"FT=";FT;"FINAL TIME,SEC"
820 PRINT#2,"DS=";DS;"AZIMUTHAL INCREMENT,DEG"
830 PRINT#2,"OM=";OM;"MAIN ROTOR SPEED,RPM"
840 PRINT#2,"AL=";AL;"INITIAL ALTITUDE,FT"
850 REM V=Aircraft Initial Speed,Kt
860 REM FT=Final Time Of Iteration, Sec
870 REM DS=Delta Azimuthal Increment For Solution Step,Deg
880 REM OM=Main Rotor Normal Rotational Speed,RPM
890 REM AL=Aircraft Initial Height Above Ground,Ft
900 INPUT#1,NB,NE,OB
910 PRINT NB;NE;OB
920 PRINT#2,"NB=";NB;"NUMBER OF BLADES PER ROTOR,ND"
930 PRINT#2,"NE=";NE;"NUMBER OF BLADE ELEMENTS,ND"
940 PRINT#2,"OB=";OB;"BLADE FLAP NAT FREQ,ND"
950 REM NB=Number Of Main Rotor Blades,ND
960 REM NE= "      " Elements Of Main Rotor & Stab Over
970 REM      Which Radial Integration Takes Place,ND
980 REM OB=ND blade 1st Flap Frequency,ND
990 INPUT#1,TK,CC,EE,FA
1000 PRINT TK;CC;EE;FA
1010 PRINT#2,"TK=";TK;"MAIN ROTOR BLADE THICKNESS,IN"
1020 PRINT#2,"CC=";CC;"NUMERICAL DAMPING RATIO,ND"
1030 PRINT#2,"EE=";EE;"CONVERGENCE FACTOR"
1040 PRINT#2,"FA=";FA;"FA=SPEED SWEEP FACTOR + OR -"
1050 REM TK=Main Rotor Blade Thickness,Inches
1060 REM cc=numerical damping ratio,nd
1070 REM "ee=convergence factor,nd"
1080 REM"FA=SPEED SWEEP FACTOR + OR -"
1090 INPUT#1,NC
1100 PRINT NC
1110 PRINT#2,"NC=";NC;"# OF ROTOR CYCLES TO RUN"
1120 REM"NC=# OF ROTOR CYCLES TO RUN"
1130 CLOSE 2
1140 CLOSE 1
1150 REM"....."
1160 REM".....DIMENSION MOTIONS....."
1170 REM"....DISPLACEMENTS,VELOCITIES & ACCEL....."
1180 REM ..2 ALLOWS FOR THE PRESENT, 1 FOR PAST.....
1190 REM ..EXCEPTION FP(4) ARE FLAP MOMENS.....
1200 REM .....
1210 DIM Z1(2),Z2(2),Z3(2),X1(2),X2(2),X3(2)
1220 DIM S1(2),S2(2),S3(2),P1(2),P2(2),P3(2)
1230 DIM B1(4,2),B2(4,2),B3(4,2),FP(4)
1240 REM"....."
1250 REM"..CHANGE UNITS AND CALC. INTERIAS....."
1260 REM"....."

```

```

1270 GA=GA/57.3:CG=CG/12:TW=TW/57.3
1280 CM=CM/12:CT=CT/12:OT=OT/57.3
1290 V=V*1.689:DS=DS/57.3:OM=OM*6.28/60:IB=WB*RM^2/3/32.2
1300 IP=WT*LL^2/3/32.2:TK=TK/12
1310 PP=PP/57.3:T0=T0/57.3:T2=T2/57.3
1320 REM"....."
1330 REM".....SET DT,FT,DC....."
1340 DT=DS/OM:FT=6.28/OM:DC=.1*DS
1350 REM"....."
1360 REM"....."
1370 REM".....INITIAL CONDITIONS....."
1380 REM"....."
1390 Z1(1)=AL:Z2(1)=0:Z3(1)=0
1400 X1(1)=0:X2(1)=-V:X3(1)=0
1410 S1(1)=0:S2(1)=OM:S3(1)=0
1420 P1(1)=PP:P2(1)=0:P3(1)=0
1430 OPEN"O",3,"out1"
1440 REM"....."
1450 REM".....CN=COUNT OF TIMES PASSING 1040....."
1460 REM".....PRINT TO OUT1 COLL:CYCLIC:AC PITCH....."
1470 REM".....STAB PITCH:LONG VEL:ALT:AV PWR:RPM....."
1480 CN=CN+1:PRINT#3,T0*57.3;T2*57.3;P1(10*57.3;OT*57.3;"T0 "; "T2 "; "P1 "; "OT"
1490 PRINT#3,X2(1)/1.689;Z1(1);PW/(I+IT+DT)*DT;04;"X2 "; "Z1 "; "PW "; "RPM"
1500 PRINT#3,X3(1);Z2(1);P2(1)*57.3;P3(1)*57.3;"X3 "; "Z2 "; "P2 "; "P3"
1510 PRINT#3,C2;C3;RP;(GA+P1(1)-C3/57.3)*57.3;"B0 "; "B1C"; " RP "; "GA+P-B1C"
1520 PRINT#3,"....."
1530 REM"....."
1540 REM".....INITIAL CONDITIONS....."
1550 REM"....."
1560 Z1(1)=AL:Z2(1)=0:Z3(1)=0
1570 X1(1)=0:X2(1)=-V:X3(1)=0
1580 S1(1)=0:S2(1)=OM:S3(1)=0
1590 P1(1)=P1(1):P2(1)=0:P3(1)=0
1600 REM"....."
1610 REM"..... INITIAL DOWNWASH OF MAIN....."
1620 REM".....ROTOR AND STAB ..... "
1630 REM"....."
1640 REM"....."
1650 REM".....BREAK UP OF MAIN ROTOR BLADE....."
1660 REM"....."
1670 DR=RM/NE
1680 REM"....."
1690 REM".....SET BLADE INITIAL CONDITIONS....."
1700 REM"....."
1710 FOR I=1 TO NB:B1(I,1)=0:B2(I,1)=0:B3(I,1)=0:NEXT I
1720 RR=0
1730 PW=0
1740 TN=0
1750 REM"....."
1760 REM".....START OF TIME ITERATION....."
1770 REM"....."
1780 FOR I=0 TO NC*FT STEP DT
1790 U=SQR(V^4+(WT/RH/3.14/RM^2)^2)
1800 U=SQR(1/2*(-V^2+U))
1810 UU=SQR(V^4+(WT/10/RH/2/CT/RT)^2)
1820 UU=SQR(1/2*(-V^2+UU))
1821 REM"MU=-X2(1)/S2(1)/RM"
1822 REM"UZ=U+X2(1)*(GA+P1(1))"
1823 REM"LA=UZ/S2(1)/RM"
1824 REM"AO=1/3+MU/2*(GA+P1(1)-LA)"
1825 REM"BO=MU/2*(GA+P1(1)-LA)"
1826 REM"WW=WT/(NB/2)/RH/CM/RM/MM/(S2(1)*RM)^2"
1827 REM"AR=ABS(HH*WW*COS(P1(1))+WW*LL*SIN(P1(1)))"

```

```

1828 REM"AB=AR/CM/MM/RM"
1829 REM"E=WW+LA/2-TW*RM/4*(1+(MU^2+2*LA^2))"
1830 REM"F=-AB*MU^2-TW*RM*MU*((GA+P1(1))/3-LA/4)"
1831 REM"F=F+(GA+P1(1))*MU*LA/2+D0*MU/MM/2"
1832 REM"D=(1/6+MU^2/4+LA^2/2)*(GA+P1(1))-LA/4"
1833 REM"CO=MU/2"
1834 REM"T2=(AO*F-BO*E)/(AO*D-BO*CO)"
1839 REM"....."
1840 REM".....TIME STEP & ELAPSED TIME....."
1850 TN=TN+DT
1860 REM"....."
1870 REM".....ALTITUDE....."
1880 O1=INT(100*Z1(1))/100
1890 REM"....."
1900 REM".....VERTICAL VELOCITY....."
1910 O2=INT(Z2(1))
1920 REM"....."
1930 REM".....LONGITUDINAL VELOCITY....."
1940 O3=INT(100*X2(1)/1.689)/100
1950 REM"....."
1960 REM".....ROTOR SPEED....."
1970 O4=INT(100*S2(1)*60/6.28)/100
1980 REM"....."
1990 REM".....AIRCRAFT PITCH....."
2000 O5=INT(100*P1(1)*57.3)/100
2010 REM"....."
2020 REM".....NUMBER OF ROTOR CYCLES....."
2030 O6=INT(100*S1(1)/6.28)/100:GOTO 2060
2040 REM"....."
2050 REM"...IF TIME >.98*NC*FT, TEST TO SEE IF CRITERION IS MET...."
2060 IF TN>.98*NC*FT THEN 2150
2070 REM"....."
2080 REM"...IF TIME <.98*NC*FT,CONTINUE TO ITERATE IN TIME....."
2090 GOTO 2440
2100 REM"....."
2110 REM"...TESTS ON PARAMETERS FOR TRIM,IF TESTS ARE PASSED CU=0....."
2120 REM"...THE FWD SPEED IS INCREMENTED & GOES TO 1060....."
2130 REM"...WHERE NEW IC ARE SET. IF TESTS FAIL GO TO 1360....."
2140 REM"....."
2150 REM"IF ABS(X2(1)+V)/100/1.689>.02 THEN 2440"
2160 IF ABS(X3(1))>.5 THEN 2440
2170 REM"IF ABS(Z1(1)-AL)/100>.02 THEN 2440"
2180 IF ABS(Z2(1))>.5 THEN 2440
2190 REM"IF ABS(P2(1)*57.3)>2 THEN 2440"
2200 REM"....."
2210 REM".....TESTS PASSED, PRINT OUT DATA TO OUT1....."
2220 REM"....PRINT COLL:CYCLIC:AIRCRAFT PITCH:STAB PITCH:...."
2230 REM".....LONG VEL:ALT:PWR:....."
2240 REM".....RPM....."
2250 PRINT#3,"criteria met"
2260 PRINT#3,T0*57.3;T2*57.3;P1(1)*57.3;OT*57.3;"T0 "; "2 "; "P1 "; "OT"
2270 PRINT#3,X2(1)/1.689;Z1(1);PW/(I+IT+DT)*DT;O4;"X2 "; "Z1 "; "PW "; "RPM"
2280 PRINT#3,X3(1);Z2(1);P2(1)*57.3;P3(1)*57.3;"X3 "; "Z2 "; "P2 "; "P3"
2290 PRINT#3,C2;C3;RP;(GA+P1(1)-C3)*57.3;"B0 "; "B1C"; "RP "; "GA+P-B1C"
2300 PRINT#3,"....."
2310 REM"....."
2320 REM".....PRINT TO SCREEN COLL:CYCLIC:AC PITCH:....."
2330 REM".....STAB PITCH:AC PITCH RATE:STRING....."
2340 PRINT"criteria met";T0*57.3;T2*57.3;P1(1)*57.3;OT*57.3;P2(1)*57.3;G$
2350 REM"....."
2360 REM".....INCREMENT FWD SPEED OR STOP IF V>160 OR <1 KT..."
2370 V=(V+FA*5*1.689)
2380 IF V > 160*1.689 THEN STOP

```



```

2390 IF V<1*1.689 THEN STOP
2400 GOTO 1560
2410 REM"....."
2420 REM"...IF ELAPSED TIME > .98*NC*FT, THEN PRINT DATA AT 1040.."
2430 REM"...& START NEW AIR SPEED....."
2440 IF TN> .98*NC*FT THEN 2451
2450 GOTO 2530
2451 PRINT#3,T0*57.3;T2*57.3;P1(1)*57.3;OT*57.3;"T0 "; "T2 "; "P1 "; "OT"
2452 PRINT#3,X2(1)/1.689;Z1(1);PW/(I+IT+DT)*DT;O4;"X2 "; "Z1 "; "PWR "; "RPM"
2453 PRINT#3,X3(1);Z2(1);P2(1)*57.3;P3(1)*57.3;O6;"X3 "; "Z2 "; "P2 "; "P3"
2454 PRINT#3,(GA+P1(1)+C3/57.3)*57.3
2455 PRINT#3,"GA+P+B1C"
2456 PRINT#3,C2;C3;RP;"B0 "; "B1C"; " RP"
2457 PRINT#3,"....."
2460 IF Z2(1) > 0 THEN T0=T0-.1/57.3
2470 IF Z2(1) < 0 THEN T0=T0+.1/57.3
2480 IF X3(1) > 0 THEN T2=T2-.2/57.3
2490 IF X3(1) < 0 THEN T2=T2+.2/57.3
2491 REM"T2=(AO*F-BO*E)/(AO*D-BO*CO)"
2500 REM"IF P2(1) > 0 THEN OT=OT+.1/57.3"
2510 REM"IF P2(1) < 0 THEN OT=OT-.1/57.3"
2511 REM"IF Z2(1)>0 THEN OT=OT-.1/57.3"
2512 REM"IF Z2(1)<0 THEN OT=OT+.1/57.3"
2513 IF X3(1)>0 THEN OT=OT+.1/57.3
2514 IF X3(1)<0 THEN OT=OT-.1/57.3
2520 GOTO 1540
2530 REM"....."
2540 REM"....REQUIRED POWER....."
2550 O7=INT(HP)
2560 REM"....."
2590 REM"....."
2600 REM".....PITCH RATE....."
2610 A6=INT(10*P2(1)*57.3/10)
2620 REM"....."
2630 REM".....SET THRUST PARALLEL TO SHAFT, FORCE PERP> TO....."
2640 REM".....SHAFT,TORQUE,VERT FORCE, HOR. FORCE,& PITCH....."
2650 REM".....EQUAL ZERO....."
2660 TX=0:TQ=0:FV=0:FX=0:MP=0
2670 REM"....."
2680 REM".....PRINT TO SCREEN....."
2690 REM"....."
2700 REM".....PRINT TO OUT....."
2710 C2=0:C3=0
2720 REM"....."
2730 REM".....START AZIMUTHAL SUMMING....."
2740 FOR J=1 TO NB
2750 REM"....."
2760 REM".....FLAP MOMENTS SET TO ZERO....."
2770 FP(J)=0
2780 REM"....."
2790 REM".....START RADIAL INTEGRATION....."
2800 FOR IK=1 TO NE
2810 REM"....."
2820 REM".....AZIMUTHAL POSITION OF JTH BLADE....."
2830 SS=S1(1)+(J-1)*6.28/NB
2840 REM"....."
2850 REM".....BLADE TOTAL PITCH....."
2860 TH=T0+TW*DR*IK+T2*SIN(SS)
2870 REM"....."
2880 REM".....COMPONENTS OF UP & UT....."
2890 REM"....."
2900 REM".....UP....."
2910 UP=-B2(J,1)*IK*DR

```

```

2911 UP=UP+X2(1)*COS(SS)*B1(J,1)
2912 UP=UP-Z2(1)*(GA+P1(1))*COS(SS)*B1(J,1)
2913 UP=UP+X2(1)*(GA+P1(1))
2914 UP=UP+Z2(1)
2915 UP=UP+P2(1)*ZR*B1(J,1)*COS(SS)
2916 UP=UP+P2(1)*(-IK*DR*COS(SS)-CG)
2917 UP=UP+U
2930 REM"....."
2940 REM".....UT....."
2950 UT=-X2(1)*SIN(SS)
2951 UT=UT+Z2(1)*(GA+P1(1))*SIN(SS)
2952 UT=UT-P2(1)*ZR*SIN(SS)
2953 UT=UT+IK*DR*S2(1)
2960 REM"....."
2970 REM".....CALC PHI=ATAN(UP/UT)....."
2980 IF UP> 0 AND UT >0 THEN PM=ATN(ABS(UP/UT))
2990 IF UP< 0 AND UT >0 THEN PM=-ATN(ABS(UP/UT))
3000 IF UP> 0 AND UT <0 THEN PM=-ATN(ABS(UP/UT))+3.14
3010 IF UP< 0 AND UT <0 THEN PM=ATN(ABS(UP/UT))+3.14
3020 REM"....."
3030 REM".....CALC. BLADE LIFT....."
3040 L=.5*RH*CM*MM*(TH-PM+EM)*(UP^2+UT^2)*DR
3050 REM"....."
3060 REM".....CALC. BLADE DRAG....."
3070 D=.5*RH*CM*(DO+K*(SIN(TH-PM+EM))^2)*(UP^2+UT^2)*DR
3080 REM"....."
3090 REM".....CALC. THRUST PARALLEL TO SHAFT....."
3100 T=L*COS(PM)-D*SIN(PM)
3110 TX=TX+T
3120 FZ=T
3130 REM"....."
3140 REM".....CALC. FORCE PERPENDICULAR TO SHAFT....."
3150 H=L*SIN(PM)+D*COS(PM)
3160 FY=-H
3170 REM"....."
3180 REM"....CALC. AIR RESISTANCE TORQUE OF ROTOR....."
3190 TQ=TQ+H*IK*DR
3200 REM"....."
3210 REM".....CALC. TOTAL VERTICAL FORCE....."
3220 FV=FV+FZ
3230 REM"....."
3240 REM".....CALC. TOTAL HORIZONTAL FORCE....."
3250 DM=FZ*B1(J,1)*COS(SS)-FY*SIN(SS)+GA*FZ+P1(1)*FZ
3260 FX=FX+DM
3270 REM"....."
3280 REM".....CALC. TOTAL AIRCRAFT PITCH MOMENT....."
3290 MP=MP+DM*(ZR-P1(1)*IK*DR*COS(SS)-IK*DR*GA*COS(SS)-IK*DR*B1(J,1))
3300 MP=MP+FZ*(-CG-ZR*GA-P1(1)*ZR-IK*DR*COS(SS))
3310 REM"....."
3320 REM".....CALC. FLAP MOMENTS....."
3330 FP(J)=FP(J)+DM*(-IK*DR*B1(J,1)*COS(SS)-IK*DR*(P1(1)+GA))-FZ*IK*DR
3340 REM"....."
3350 REM"..UP & UT FOR BLADE #1 @ TIP, LATER TO CALC ANGLE OF ATTACK..."
3360 IF J=1 AND IK = NE THEN AX=INT(100*UP)/100
3370 IF J=1 AND IK = NE THEN BB=INT(100*UT)/100
3380 NEXT IK
3390 REM"....."
3400 REM".....END OF RADIAL INTEGRATION....."
3410 REM"....."
3420 REM".....FLAP ACCEL. ONE TIME STEP PAST....."
3430 B3(J,1)=-OB^2*S2(1)^2*B1(J,1)+FP(J)/IB
3440 B3(J,1)=B3(J,1)-.5*RH*CM*RM*RM*(RM*B2(J,1)^2*B2(J,1)/ABS(B2(J,1)+.1)/IB
3450 B3(J,1)=B3(J,1)-2*CC*S2(1)*B2(J,1)

```

```

3460 REM".....PRESENT FLAP ANGLE....."
3470 REM".....PRESENT FLAP ANGLE....."
3480 B1(J,2)=B1(J,1)+B2(J,1)*DT+B3(J,1)*DT^2/2
3490 REM".....PRESENT FLAP VELOCITIES....."
3500 REM".....PRESENT FLAP VELOCITIES....."
3510 B2(J,2)=B2(J,1)+B3(J,1)*DT
3520 REM"....."
3530 REM"...SET NEXT FLAP ANGLES & VELOCITIES ONE STEP BEHIND...."
3540 B1(J,1)=B1(J,2):B2(J,1)=B2(J,2)
3550 REM"....."
3560 REM".....CALC. CONING ANGLE....."
3570 C2=C2+B1(J,1)
3580 REM"....."
3590 REM".....CALC. DISC PLANE LONG. TILT....."
3600 C3=C3+B1(J,1)*COS(SS)
3610 NEXT J
3620 REM"....."
3630 REM".....END OF AZIMUTHAL SUMMING....."
3640 REM"....."
3650 REM"...INERTIA TORQUE & AIR RESISTANCE TORQUE....."
3660 RP=NB*IB*S3(1)+TQ
3670 REM"....."
3680 REM".....DRAG & NUMERICAL DAMPING TORQUE....."
3690 REM"....."
3700 REM".....TOTAL TORQUE....."
3710 RP+RP
3720 REM"....."
3730 REM".....TOTAL REQUIRED POWER....."
3740 RP=RP*S2(1)/550
3750 REM"....."
3760 REM"....."
3770 REM"...SET HP FOR CONSTANT RPM....."
3780 HP=RP
3790 REM"....."
3800 REM".....SUMMING POWERS....."
3810 PW=PW+HP
3820 RR=RR+RP
3830 REM"....."
3840 REM".....AZIMUTHAL ACCEL....."
3850 S3(1)=-TQ+HP*550/S2(1)
3860 S3(1)=S3(1)/NB/IB
3870 REM"....."
3880 REM"....."
3890 REM".....PRINT AZIMUTHAL ACCEL & POWERS TO SCREEN & OUT..."
3900 REM"....."
3910 REM".....STAB UP & UT....."
3920 PU=+X2(1)*SIN(P1(1))+Z2(1)-P2(1)*XT+UU
3930 TU=-X2(1)-P2(1)*ZT
3940 REM"....."
3950 REM".....STAB PHI....."
3960 IF PU > 0 AND TU>0 THEN PT=ATN(ABS(PU/TU))
3970 IF PU < 0 AND TU>0 THEN PT=-ATN(ABS(PU/TU))
3980 IF PU > 0 AND TU < 0 THEN PT=-ATN(ABS(PU/TU))+3.14
3990 IF PU < 0 AND TU < 0 THEN PT=ATN(ABS(PU.TU))+3.14
4000 REM"....."
4010 REM"...STAB LIFT,DRAG,THRUST PERP TO STAB,STAB FORCE ...."
4020 REM"...PARALLEL TO STAB....."
4030 TL=.5*RH*CT*TM*(OT-PT+ET)*(PU^2+TU)*RT
4040 TD=.5*RH*CT*(OD+KT*(SIN(OT-PT+ET))^2*(PU^2+TU^2)*RT
4050 TT=TL*COS(PT)-TD*SIN(PT)
4060 HT=TL*SIN(PT)+TD*COS(PT)
4070 REM"....."
4080 REM".....AIRCRAFT VERT ACCEL....."

```

```

4090 Z3(1)=FV
4100 Z3(1)=Z3(1)+2*TT-2*HT*P1(1)
4110 Z3(1)=Z3(1)-WT
4120 Z3(1)=Z3(1)-.5*RH*LL*WW*Z2(1)^2*Z2(1)/ABS(Z2(1)+.1)
4130 Z3(1)=Z3(1)
4140 Z3(1)=Z3(1)*32.3/WT
4150 REM"....."
4160 REM".....AIRCRAFT HORIZONTAL ACCEL....."
4170 X3(1)=FX
4180 X3(1)=X3(1)+2*TT*P1(1)+2*HT
4190 AA=.5*RH*WW*ABS(HH*COS(P1(1))+LL*SIN(P1(1)))*X2(1)^3/ABS(X2(1))
4200 X3(1)=X3(1)-AA
4210 X3(1)=X3(1)*32.2/WT
4220 REM"....."
4230 REM".....AIRCRAFT PITCH ACCEL....."
4240 P3(1)=MP
4250 P3(1)=P3(1)-2*TT*XT+2*(HT+TT*P1(1))*ZT
4260 P3(1)=P3(1)-.5*RH*LL*WW*LL*(LL*P2(1))^2*P2(1)/ABS(P2(1)+.1)
4270 PE(1)=P3(1)-.5*RH*LL*WW*LL*P1(1)*(X2(1))^2
4280 P3(1)=P3(1)/IP-2*CC*S2(1)*P2(1)
4290 REM"....."
4300 REM"....DETERMINE SIGN OF DOWNWASH VELOCITY OF MAIN...."
4310 REM"....ROTOR & TAIL....."
4320 U=SQR(V^4+(TX/RH/3.14/RM^2)^2)
4330 U=SQR(1/2*(-V^2+U))
4340 IF TX>0 THEN U=U
4350 IF TX<0 THEN U=-U
4360 UU=SQR(V^4+(TT/RH/2/CT/RT)^2)
4370 UU=SQR(1/2*(-V^2+UU))
4380 IF TT>0 THEN UU=UU
4390 IF TT<0 THEN UU=-UU
4400 REM"....."
4410 REM"....CALC PRESENT VERT,HOR,AZIM,PITCH DISPL & VEL...."
4420 Z1(2)=Z1(1)+Z2(1)*DT+Z3(1)*DT^2/2
4430 Z2(2)=Z2(1)+Z3(1)*DT
4440 X1(2)=X1(1)+X2(1)*DT+X3(1)*DT^2/2
4450 X2(2)=X2(1)+X3(1)*DT
4460 S1(2)=S1(1)+S2(1)*DT+S3(1)*DT^2/2
4470 S2(2)=S2(1)+S3(1)*DT
4480 P1(2)=P1(1)+P2(1)*DT+P3(1)*DT^2/2
4490 P2(2)=P2(1)+P3(1)*DT
4500 REM"....."
4510 REM"....SET PP=PRESENT AIRCRAFT PITCH....."
4520 PP=P1(2)
4530 REM"....."
4540 REM"....."
4550 REM"....SET DIPL & VEL ONE STEP IN PAST....."
4560 Z1(1)=Z1(2):Z2(1)=Z2(2)
4570 X1(1)=X1(2):X2(1)=X2(2)
4580 S1(1)=S1(2):S2(1)=S2(2)
4590 P1(1)=P1(2):P2(1)=P2(2)
4600 REM"....."
4610 REM"....COLLECTIVE PITCH....."
4620 A7=INT(10*T0*57.3)/10
4630 REM"....."
4640 REM"....LONG CYCLIC PITCH....."
4650 A8=INT(10*T2*57.3)/10
4660 REM"....."
4670 REM"....STAB PITCH ANGLE....."
4680 A9=INT(10*OT*57.3)/10
4690 REM"....."
4700 REM"....ELAPSED TIME....."
4710 C1=INT(1000*TN)/1000

```

```

4720 REM"....."
4730 REM".....CONING....."
4740 C2=INT(10*C2/NB*57.3)/10
4750 REM"....."
4760 REM".....DISC PLANE LONG TILT....."
4770 C3=INT(10*C3/NB*57.3)/10
4780 REM"....."
4790 REM"....."
4800 PRINT T0*57.3;T2*57.3;P1(1)*57.3;OT*57.3;"T0 "; "T2 "; "P1 "; "OT"
4810 PRINT X2(1)/1.689;Z1(1);PW/(I+IT+DT)*DT;O4;"X2 "; "Z1 "; "PWR "; "RPM"
4820 PRINT X3(1);Z2(1);P2(1)*57.3;P3(1)*57.3;O6;"X3 "; "Z2 "; "P2 "; "P3"
4830 PRINT (GA+P1(1)+C3/57.3)*57.3;"GA+P+B1C"
4840 PRINT C2;C3;RP;"B0 "; "B1C";" RP":PRINT ".....":GOTO 5020
4850 PRINT FV;TT;HT;TL;TD;PT;FX;MP;"FV "; "TT "; "HT "; "TL "; "TD "; "PT "; "FX "; "MP"
4860 PRINT PU;TU;UU;T;TX;PM;"PU "; "TU "; "UU "; "T "; "TX "; "PM"
4870 PRINT UP;UT;U;TQ;AC;S3(1);"UP "; "UT "; "U "; "TQ "; "AC "; "S3"
4880 PRINT "....."
4890 PRINT#3,T0*57.3;T2*57.3;P1(1)*57.3;OT*57.3;"T0 "; "T2 "; "P1 "; "OT"
4900 PRINT#3,X2(1)/1.689;Z1(1);PW/(I+IT+DT)*DT;O4;"X2 "; "Z1 "; "PWR "; "RPM"
4910 PRINT#3,X3(1);Z2(1);P2(1)*57.3;P3(1)*57.3;O6;"X3 "; "Z2 "; "P2 "; "P3"
4920 PRINT#3,(GA+P1(1)+C3/57.3)*57.3
4930 PRINT#3,"GA+P+B1C"
4940 PRINT#3,C2;C3;RP;"B0 "; "B1C";" RP":REM"GOTO 5010"
4950 PRINT#3,FV;TT;HT;TL;TD;PT;FX
4960 PRINT#3,"FV "; "TT "; "HT "; "TL "; "TD "; "PT "; "FX "
4970 PRINT#3,PU;TU;UU;MP;UP;UT;U
4980 PRINT#3,"PU "; "TU "; "UU "; "MP "; "UP "; "UT "; "U"
4990 PRINT#3,T;TQ;AC;S3(1);TX;PM
5000 PRINT#3,"T "; "TQ "; "AC "; "S3 "; "TX "; "PM"
5010 PRINT#3,"....."
5020 NEXT I
5030 REM"....."
5040 REM".....END OF TIME LOOP....."
5050 REM
5060 REM
5070 END

```

INTENTIONALLY LEFT BLANK.

LIST OF ABBREVIATIONS AND SYMBOLS

$X(2)$	=	displacement at present time
$X(1)$	=	displacement from previous time step
$\frac{dX(1)}{dt}$	=	velocity from previous time step
$\frac{d^2X(1)}{dt^2}$	=	acceleration from previous time step
δt	=	time step
$\frac{dD}{dr}$	=	drag per blade span
$\frac{dL}{dr}$	=	lift per blade span
ρ	=	air density
C	=	main rotor blade chord
a	=	main rotor blade lift curve slope
α	=	main rotor blade angle of attack
U_p	=	air velocity parallel to main rotor shaft
U_t	=	air velocity perpendicular to main rotor shaft
C_D	=	average main rotor blade drag coefficient

INTENTIONALLY LEFT BLANK.

<u>NO. OF COPIES</u>	<u>ORGANIZATION</u>
2	DEFENSE TECHNICAL INFO CTR ATTN DTIC DDA 8725 JOHN J KINGMAN RD STE 0944 FT BELVOIR VA 22060-6218

1	DIRECTOR US ARMY RESEARCH LAB ATTN AMSRL OP SD TA 2800 POWDER MILL RD ADELPHI MD 20783-1145
---	---

3	DIRECTOR US ARMY RESEARCH LAB ATTN AMSRL OP SD TL 2800 POWDER MILL RD ADELPHI MD 20783-1145
---	---

1	DIRECTOR US ARMY RESEARCH LAB ATTN AMSRL OP SD TP 2800 POWDER MILL RD ADELPHI MD 20783-1145
---	---

ABERDEEN PROVING GROUND

2	DIR USARL ATTN AMSRL OP AP L (305)
---	---------------------------------------

<u>NO. OF COPIES</u>	<u>ORGANIZATION</u>
1	OSD OUSD AT STRT TAC SYS ATTN DR SCHNEITER 3090 DEFNS PENTAGON RM 3E130 WASHINGTON DC 20301-3090
1	ODDRE AT ACQUISITION AND TECH ATTN DR GONTAREK 3080 DEFENSE PENTAGON WASHINGTON DC 20310-3080
1	ASST SECY ARMY RESEARCH DEVELOPMENT ACQUISITION ATTN SARD ZD RM 2E673 103 ARMY PENTAGON WASHINGTON DC 20310-0103
1	ASST SECY ARMY RESEARCH DEVELOPMENT ACQUISITION ATTN SARD ZP RM 2E661 103 ARMY PENTAGON WASHINGTON DC 20310-0103
1	ASST SECY ARMY RESEARCH DEVELOPMENT ACQUISITION ATTN SARD ZS RM 3E448 103 ARMY PENTAGON WASHINGTON DC 20310-0103
1	ASST SECY ARMY RESEARCH DEVELOPMENT ACQUISITION ATTN SARD ZT RM 3E374 103 ARMY PENTAGON WASHINGTON DC 20310-0103
1	UNDER SEC OF THE ARMY ATTN DUSA OR RM 2E660 102 ARMY PENTAGON WASHINGTON DC 20310-0102
1	ASST DEP CHIEF OF STAFF OPERATIONS AND PLANS ATTN DAMO FDZ RM 3A522 460 ARMY PENTAGON WASHINGTON DC 20310-0460
1	DEPUTY CHIEF OF STAFF OPERATIONS AND PLANS ATTN DAMO SW RM 3C630 400 ARMY PENTAGON WASHINGTON DC 20310-0400

<u>NO. OF COPIES</u>	<u>ORGANIZATION</u>
1	ARMY RESEARCH LABORATORY ATTN AMSRL SL DR WADE WSMR NM 88002-5513
1	ARMY RESEARCH LABORATORY ATTN AMSRL SL E MR MARES WSMR NM 88002-5513
1	ARMY TRADOC ANL CTR ATTN ATRC W MR KEINTZ WSMR NM 88002-5502
1	ARMY TRNG & DOCTRINE CMND ATTN ATCD B FT MONROE VA 23651
	<u>ABERDEEN PROVING GROUND</u>
1	CDR USATECOM ATTN: AMSTE-TA
2	DIR USAMSAA ATTN: AMXSY-ST AMXSY-D
4	DIR USARL ATTN: AMSRL-SL, J WADE (433) AMSRL-SL-I, M STARKS (433) AMSRL-SL-C, W HUGHES (E3331) AMSRL-SL-B, P DEITZ (328)

NO. OF
COPIES ORGANIZATION

- 2 COMMANDER
USA ATCOM
ATTN SFAE AV AAH LBA
LTC R GUNNING
J GRACZAK
4300 GOODFELLOW BLVD
ST LOUIS MO 63120-1798
- 1 MCDONNELL DOUGLAS HEL SYS
DEPT MGR LONGBOW T&E
ATTN B GUPTA
MS 530 B334
5000 E MCDOWELL RD
MESA AZ 85215-9797
- 1 COMMANDER
USA ATCOM
ATTN AMSAT R ESC
G KOVACS
4300 GOODFELLOW BLVD
ST LOUIS MO 63120-1798
- 1 US ARMY AVIATION APPL TECH DIR
ATTN AMSAT R TV
G BIROCCO
FT EUSTIS VA 23604-5577

ABERDEEN PROVING GROUND

- 6 DIR, USARL
ATTN: AMSRL-SL-BA,
J. MORRISSEY
S. POLYAK
K. KIM
J. FRIES
AMSRL-SL-I,
D. HASKELL
M. VOGEL
- 2 DIR, US AMSAA
ATTN: AMXSY-CB,
W. FERGUSON
AMXSY-D,
A. LAGRANGE

INTENTIONALLY LEFT BLANK.

USER EVALUATION SHEET/CHANGE OF ADDRESS

This Laboratory undertakes a continuing effort to improve the quality of the reports it publishes. Your comments/answers to the items/questions below will aid us in our efforts.

1. ARL Report Number/Author ARL-MR-317 (Fries) Date of Report June 1996

2. Date Report Received _____

3. Does this report satisfy a need? (Comment on purpose, related project, or other area of interest for which the report will be used.) _____

4. Specifically, how is the report being used? (Information source, design data, procedure, source of ideas, etc.) _____

5. Has the information in this report led to any quantitative savings as far as man-hours or dollars saved, operating costs avoided, or efficiencies achieved, etc? If so, please elaborate. _____

6. General Comments. What do you think should be changed to improve future reports? (Indicate changes to organization, technical content, format, etc.) _____

CURRENT
ADDRESS

Organization

Name

Street or P.O. Box No.

City, State, Zip Code

7. If indicating a Change of Address or Address Correction, please provide the Current or Correct address above and the Old or Incorrect address below.

OLD
ADDRESS

Organization

Name

Street or P.O. Box No.

City, State, Zip Code

(Remove this sheet, fold as indicated, tape closed, and mail.)
(DO NOT STAPLE)

DEPARTMENT OF THE ARMY

OFFICIAL BUSINESS

BUSINESS REPLY MAIL
FIRST CLASS PERMIT NO 0001,APG,MD

POSTAGE WILL BE PAID BY ADDRESSEE

DIRECTOR
U.S. ARMY RESEARCH LABORATORY
ATTN: AMSRL-SL-BA
ABERDEEN PROVING GROUND, MD 21005-5068



NO POSTAGE
NECESSARY
IF MAILED
IN THE
UNITED STATES

